The problem then reduces to calculating the following:

Y = (A0i*X0 + A1i*X1 + A2i*X2 + A3i*X3) + (A0f*X0 + A1f*X1 + A2f*X2 + A3f*X3)

This is like calculating two filter banks. The above problem is coded in the example below:

```
; Assume:        X0,X1,X2,X3 = Q15 (-1    range    0.999053955)
;                Y = Q10 (-32   range   +31.99902344)
;  Ymin-max = 2.391456 + 0.0235045 + 0.000329758 + 34.3392345
;                = +/- 36.75452476
;                Sat      = 06000h
;                Round    = 08000h

        SETC    OVM     ; Enable saturation.
        SETC    SXM     ; Enable sign extension.
        SPM     3       ; Set shift mode = -6
        LT      A0f
        MPY     X0      ; P = A0f*X0
        LTP     A1f     ; ACC = A0f*X0
        MPY     X1      ; P = A1f*X1
        LTA     A2f     ; ACC = ACC + A1F*X1
        MPY     X2      ; P = A2f*X2
        LTA     A3f     ; ACC = ACC + A2f*X2
        MPY     X3      ; P = A3f*X3
        LTA     A0i     ; ACC = ACC + A3f*X3
        SPM     0
        SACH    Temp,6  ; On C5X replace by BSAR 9
        LAC     Temp,1  ; ACC = ACC/512
        ; instruction.
        MPY     X0      ; P = A0i*X0
        LTA     A1i     ; ACC = ACC + A0i*X0
        MPY     X1      ; P = A1i*X1
        LTA     A2i     ; ACC = ACC + A1i*X1
        MPY     X2      ; P = A2i*X2
        LTA     A3i     ; ACC = ACC + A2i*X2
        MPY     X3      ; P = A3i*X3
        APAC            ; ACC = ACC + A3i*X3
        ADDS    Round   ; Round result.
        ADDH    Sat     ; Saturate Y to Q10 value
        SUBH    Sat
        SUBH    Sat
        ADDH    Sat
        SACH    Y,1     ; Y = Q10 number.

; Cycles = 13 + 4n cycles (n = number of taps).

; Note: If saturation is not required, Cycles = 8 + 4n cycles
```

*Figure 1.*

If the number of taps is greater then 6, then a RPT loop can be used for each bank and the effective cycles/tap can be approximately 2.

The above technique is almost equivalent to a floating-point notation with a 4-bit exponent and a 16-bit mantissa.